

Appendix 1

A.T. Tredennick, A.R. Kleinhesselink, B. Taylor & P.B. Adler

“Consistent ecosystem functional response across precipitation extremes in a sagebrush steppe”

PeerJ

Section A1.1 Estimating ANPP

We used a radiometer to nondestructively estimate aboveground net primary productivity. Our approach relies on relating greenness in a plot to aboveground biomass. In each year we recorded ground reflectances at four bands, two associated with the red spectrum and two associated with the near-infrared spectrum (Table A1-1). We took four readings per plot that were averaged for each band. Bands 1 and 3 correspond to wavelengths collected by the MODIS satellite and bands 2 and 4 correspond to wavelengths collected by the AVHRR satellite.

Table A1-1 Radiometer specifications.

Band number	Spectrum name	Center wavelength	Corresponding satellite
1	red	626 nm	AVHRR
2	red	652 nm	MODIS
3	near-infrared	875 nm	AVHRR
4	near-infrared	859 nm	MODIS

Using the RED and NIR reflectance values, we calculate the normalized difference vegetation index (NDVI) for each plot based on both AVHRR- and MODIS-based wavelengths. We calculated NDVI as:

$$\text{NDVI}_{\text{AVHRR}} = \frac{b_3 - b_1}{b_3 + b_1} \quad (1)$$

$$\text{NDVI}_{\text{MODIS}} = \frac{b_4 - b_2}{b_4 + b_2} \quad (2)$$

where b_x refers to band x ($x = 1,2,3,4$) in Table A1-1.

To convert plot NDVI to biomass, we regressed known biomass values from calibration plots against NDVI calculate for those plots. Calibration plots were located near our experiment plots, and each year we located a new set of 10 plots in which we clipped all aboveground biomass, dried it to a constant weight at 60° C, and the weighed. We used these biomass values to estimate regression parameters for both AVHRR- and MODIS-based NDVI. We assessed model fit using

17 R^2 and, for each year, we used the regression parameters associated with the best fit model to
 18 estimate biomass in the experimental plots based on their NDVI values (Table A1-2). R code for
 19 this procedure is in the file “calibrate_radiometer_by_year.R” in the supplemental code set.

Table A1-2 Details of regression models used to estimate biomass each year.

Year	Intercept	NDVI Slope	R^2	Min(biomass)	Max(biomass)	Algorithm	NA
2012	9.03	144.23	0.59	8.57	41.42	15	MODIS
2013	1.44	111.39	0.39	8.63	77.62	15	MODIS
2014	16.31	222.38	0.63	14.61	62.30	15	MODIS
2015	-8.89	210.31	0.21	44.72	129.03	12	AVHRR
2016	14.15	493.85	0.72	50.16	163.70	16	MODIS
2017	21.67	205.64	0.57			39	MODIS

Appendix 2

A.T. Tredennick, A.R. Kleinhesselink, B. Taylor & P.B. Adler

“Consistent ecosystem functional response across precipitation extremes in a sagebrush steppe”

PeerJ

Section A2.1 Random Slopes, Random Intercepts Model

Section A2.1.1 Model Description

Our hierarchical Bayesian model allows us to test for differences among treatments in the relationship between ANPP and soil moisture, and allowed us to account for the non-independence of observations through time within a plot. Treatment differences are modeled as fixed effects, which are modified by plot-level random effects. In what follows, \mathbf{X} is the fixed effects design matrix including:

1. a column of 1s for intercepts
2. a column of continuous values for the scaled volumetric water content for each observation
3. a column of 0s or 1s indicating whether the observation is from a drought treatment
4. a column of 0s or 1s indicating whether the observation is from an irrigation treatment
5. a column of continuous values for the interaction between scaled volumetric water content and the drought treatment indicator (column 2)
6. a column of continuous values for the interaction between scaled volumetric water content and the irrigation treatment indicator (column 3)

The first seven rows of the fixed effects design matrix (six control plots and one drought plot) is as follows:

##	Int	WVC	Drought	Irrig	Drought:WVC	Irrig:WVC
## 1	1	-1.543513	0	0	0.000000	0
## 2	1	-1.543513	0	0	0.000000	0
## 3	1	-1.543513	0	0	0.000000	0
## 4	1	-1.543513	0	0	0.000000	0
## 5	1	-1.543513	0	0	0.000000	0
## 6	1	-1.543513	0	0	0.000000	0
## 7	1	-2.063694	1	0	-2.063694	0

Note that within a year, all plots within a treatment share the same value of volumetric water content. This is because we could not monitor soil moisture in each plot, and instead used sparse observations to model average soil moisture in each treatment in each year (see main text). Using this design matrix, we can estimate six fixed effects (β s):

1. the intercept of the soil moisture-ANPP relationship for control plots
2. the slope of the soil moisture-ANPP relationship for control plots
3. the intercept offset for drought plots
4. the intercept offset for irrigation plots
5. the slope offset for drought plots
6. the slope offset for irrigation plots

We are particularly interested in the slope offsets, because these allow us to test whether the slopes for drought or irrigation are different from the control slope. If the slope offset is different from zero, this indicates the slopes are different. We assess whether the slope offsets are different from zero by calculating the probability that the posterior distribution is less or greater than zero (one-tailed tests). If the probability is greater than 0.95, then there is strong evidence that the slope offset is different from zero, and thus different from the control treatment slope.

To account for the fact that observations within plots through time are not independent, we include random effects that modify the fixed effects in each plot. We model these random effects (γ s) as offsets drawn from a multivariate normal distribution with mean 0 and a variance-covariance matrix (Σ) that includes a covariance between the intercept and slope offsets. We implement this random effects structure by including a random effects design matrix (\mathbf{Z}) with a column of 1s for intercept offsets and a column of continuous values for the volumetric water content for each observation.

Lastly, to account for unknown variation across years, we include random year effects. These year effects (η s) act as offsets on the intercept.

Putting it all together, our model is defined mathematically as follows, where i denotes observation, j denotes plot, and t denotes year. We assume the observations are conditionally Gaussian,

$$\mathbf{y} \sim \text{Normal}(\boldsymbol{\mu}, \sigma^2), \quad (\text{A2.1})$$

where $\boldsymbol{\mu}$ is the deterministic expectation from the regression model,

$$\mu_i = \beta \mathbf{x}_i + \gamma_{j(i)} \mathbf{z}_i + \eta_t. \quad (\text{A2.2})$$

All fixed effect β s were drawn from normally distributed priors with mean 0 and standard deviation 5: $\beta \sim \text{Normal}(0, 5)$. γ random effects were drawn from a multivariate prior centered on zero with a shared variance covariance matrix:

$$\boldsymbol{\gamma} \sim \text{MVN}(0, \boldsymbol{\Sigma}), \quad (\text{A2.3})$$

$$\boldsymbol{\Sigma} = \sigma_{plt}^2 \mathbf{R}, \quad (\text{A2.4})$$

$$\sigma_{plt} \sim \text{Cauchy}(0, 2.5), \quad (\text{A2.5})$$

$$\mathbf{R} \sim \text{LKJ}(2.0) \quad (\text{A2.6})$$

61 The random year effects ($\boldsymbol{\eta}$) are modeled as intercept offsets centered on zero with a shared
 62 variance (σ_{yr}): $\boldsymbol{\gamma} \sim \text{Normal}(0, \sigma_{yr})$.

63 The Bayesian posterior distribution of our model can be expressed as:

$$[\boldsymbol{\beta}, \boldsymbol{\gamma}, \boldsymbol{\eta}, \sigma_{yr}, \sigma_{plt}, \mathbf{R}, \sigma | \mathbf{y}] \propto \left(\prod_{i=1}^n [y_i | \boldsymbol{\beta}, \boldsymbol{\gamma}, \boldsymbol{\eta}, \sigma] \right) \left(\prod_{j=1}^J [\gamma_j | \sigma_{plt}, \mathbf{R}] \right) \quad (\text{A2.7})$$

$$\times \left(\prod_{t=1}^T [\eta_t | \sigma_{yr}] \right) [\boldsymbol{\beta}] [\sigma_{plt}] [\mathbf{R}] [\sigma_{yr}] [\sigma] \quad (\text{A2.8})$$

64 We fit the model using MCMC as implemented in the software Stan (Stan Development Team 2016).
 65 Our Stan code is below. All code necessary to reproduce our results has been archived on Figshare
 66 ([link here](#)) and released on GitHub (https://github.com/atredennick/usses_water/releases/v0.1).

```

data {
  int<lower=0> Npreds;           # number of covariates, including intercept
  int<lower=0> Npreds2;          # number of random effect covariates
  int<lower=0> Nplots;           # number of plots
  int<lower=0> Ntreats;          # number of treatments
  int<lower=0> Nobs;             # number of observations
  int<lower=0> Nyears;           # number of years
  vector[Nobs] y;               # vector of observations
  row_vector[Npreds] x[Nobs];   # design matrix for fixed effects
  row_vector[Npreds2] z[Nobs];  # simple design matrix for random effects
  int plot_id[Nobs];            # vector of plot ids
  int treat_id[Nobs];           # vector of treatment ids
  int year_id[Nobs];            # vector of year ids
}

parameters {
  vector[Npreds] beta;          # overall coefficients
  vector[Nyears] year_off;      # vector of year effects
  cholesky_factor_corr[Npreds2] L_u; # cholesky factor of plot random effect corr matrix
  vector[Npreds2] beta_plot[Nplots]; # plot level random effects
  vector<lower=0>[Npreds2] sigma_u; # plot random effect std. dev.
  real<lower=0> sd_y;            # treatment-level observation std. dev.
  real<lower=0> sigma_year;      # year std. dev. hyperprior
}

transformed parameters {
  vector[Nobs] yhat;            # vector of expected values
  vector[Npreds2] u[Nplots];    # transformed plot random effects
  matrix[Npreds2,Npreds2] Sigma_u; # plot ranef cov matrix

  Sigma_u = diag_pre_multiply(sigma_u, L_u); # plot-level covariance matrix
  for(j in 1:Nplots)
    u[j] = Sigma_u * beta_plot[j]; # plot random intercepts and slopes

  # regression model for expected values (one for each plot-year)
  for (i in 1:Nobs)
    yhat[i] = x[i]*beta + z[i]*u[plot_id[i]] + year_off[year_id[i]];

```

```

}

model {
  ##### PRIORS
  sigma_u ~ cauchy(0,2.5)
  sigma_year ~ cauchy(0,2.5)
  year_off ~ normal(0,sigma_year); # priors on year effects, shared variance
  beta ~ normal(0,5); # priors on treatment coefficients
  L_u ~ lkj_corr_cholesky(2.0); # prior on the cholesky factor which controls the
                                # correlation between plot level treatment effects

  for(i in 1:Nplots)
    beta_plot[i] ~ normal(0,1); # plot-level coefficients for intercept and slope

  ##### LIKELIHOOD
  for(i in 1:Nobs)
    y[i] ~ normal(yhat[i], sd_y); # observations vary normally around expected values
}

generated quantities{
  corr_matrix[Npreds2] R = multiply_lower_tri_self_transpose(L_u);
  cov_matrix[Npreds2] V = quad_form_diag(R,sigma_u);
}

```




Figure A2-1 Scatterplot, on the arithmetic scale, of the data and model estimates shown as solid lines. Model estimates come from treatment level coefficients (colored lines).




Figure A2-2 Prior (dashed lines) and posterior (solid line) distributions of intercepts and slopes for each treatment. The slope represents the main effect of soil moisture on $\log(\text{ANPP})$. The red line marks 0. Shrinkage of the posterior distribution and/or changes in the mean indicate the data informed model parameters beyond the information contained in the prior for all coefficients.




Figure A2-3 Posterior distributions of random year effects (intercept offsets). Kernel bandwidths of posterior densities were adjusted by a factor of 5 to smooth the density curves for visual clarity.

69 **References**

- 70 Stan Development Team. 2016. Stan: A C++ Library for Probability and Sampling, Version 2.14.1.

Appendix 3

A.T. Tredennick, A.R. Kleinhesselink, B. Taylor & P.B. Adler

"Consistent ecosystem functional response across precipitation extremes in a sagebrush steppe"

PeerJ

Section A3.1 Characterizing Extreme Precipitation Amounts

Following the proposed methods of Lemoine et al. (2016), we calculated quantiles from the empirical distribution of growing season precipitation at Dubios, ID. We chose the 1% quantile to be indicative of extreme dry conditions (drought) and the 99% quantile to be indicative of extreme wet conditions (irrigation). The data consist of 91 yearly records, which we assume are approximately normally distributed. The R code below shows our procedure, and Fig. A3-1 shows the results.

```
library(ggplot2)
weather      <- read.csv("../data/weather/ClimateIPM.csv")
mean_ppt     <- mean(weather$ppt1)
quants_ppt   <- quantile(weather$ppt1, probs = c(0.01, 0.99))
quants_ppt[1]/mean_ppt*100 # percent of mean ppt for drought
```

```
##          1%
## 46.96351
```

```
quants_ppt[2]/mean_ppt*100 # percent of mean ppt for irrigation
```

```
##          99%
## 177.727
```

```
ggplot(weather, aes(x=ppt1))+
  geom_histogram(bins=20, color="dodgerblue", fill="dodgerblue", aes(y=..density..))+
  geom_line(stat="density", color="blue")+
  geom_vline(aes(xintercept=quants_ppt[1]), linetype=2)+
  geom_vline(aes(xintercept=quants_ppt[2]), linetype=2)+
  ylab("Density")+
  xlab("Growing Season Precipitation (mm)")+
  theme_bw()+
  theme(panel.grid.minor = element_blank())
```




Figure A3-1 Density of the empirical distribution of growing season precipitation at Dubois, ID. Dashed vertical lines show the 1% and 99% quantiles, assuming a normal distribution.

16 **References**

- 17 Lemoine, N. P., J. Sheffield, J. S. Dukes, A. K. Knapp, and M. D. Smith. 2016. Terrestrial
18 Precipitation Analysis (TPA): A resource for characterizing long-term precipitation regimes and
19 extremes. *Methods in Ecology and Evolution* 7:1396–1401.