# Appendix 2

A.T. Tredennick, A.R. Kleinhesselink, B. Taylor & P.B. Adler

"Consistent ecosystem functional response across precipitation extremes in a sagebrush steppe"

*PeerJ*

## Section A2.1 Random Slopes, Random Intercepts Model

### Section A2.1.1 Model Description

Our hierarchical Bayesian model allows us to test for differences among treatments in the relationship between ANPP and soil moisture, and allowed us to account for the non-independence of observations through time within a plot. Treatment differences are modeled as fixed effects, which are modified by plot-level random effects. In what follows, $\mathbf{X}$ is the fixed effects design matrix including:

1. a column of 1s for intercepts
2. a column of continuous values for the scaled volumetric water content for each observation
3. a column of 0s or 1s indicating whether the observation is from a drought treatment
4. a column of 0s or 1s indicating whether the observation is from an irrigation treatment
5. a column of continuous values for the interaction between scaled volumetric water content and the drought treatment indicator (column 2)
6. a column of continuous values for the interaction between scaled volumetric water content and the irrigation treatment indicator (column 3)

The first seven rows of the fixed effects design matrix (six control plots and one drought plot) is as follows:

```
##   Int       VWC Drought Irrig Drought:VWC Irrig:VWC
## 1   1 -1.543513       0     0    0.000000         0
## 2   1 -1.543513       0     0    0.000000         0
## 3   1 -1.543513       0     0    0.000000         0
## 4   1 -1.543513       0     0    0.000000         0
## 5   1 -1.543513       0     0    0.000000         0
## 6   1 -1.543513       0     0    0.000000         0
## 7   1 -2.063694       1     0   -2.063694         0
```

Note that within a year, all plots within a treatment share the same value of volumetric water content. This is because we could not monitor soil moisture in each plot, and instead used sparse observations to model average soil moisture in each treatment in each year (see main text). Using this design matrix, we can estimate six fixed effects ($\beta$s):

1. the intercept of the soil moisture-ANPP relationship for control plots
2. the slope of the soil moisture-ANPP relationship for control plots
3. the intercept offset for drought plots
4. the intercept offset for irrigation plots
5. the slope offset for drought plots
6. the slope offset for irrigation plots

We are particularly interested in the slope offsets, because these allow us to test whether the slopes for drought or irrigation are different from the control slope. If the slope offset is different from zero, this indicates the slopes are different. We assess whether the slope offsets are different from zero by calculating the probability that the posterior distribution is less or greater than zero (one-tailed tests). If the probability is greater than 0.95, then there is strong evidence that the slope offset is different from zero, and thus different from the control treatment slope.

To account for the fact that observations within plots through time are not independent, we include random effects that modify the fixed effects in each plot. We model these random effects ($\gamma$s) as offsets drawn from a multivariate normal distribution with mean 0 and a variance-covariance matrix ($\Sigma$) that includes a covariance between the intercept and slope offsets. We implement this random effects structure by including a random effects design matrix (**Z**) with a column of 1s for intercept offsets and a column of continuous values for the volumetric water content for each observation.

Lastly, to account for unkown variation across years, we include random year effects. These year effects ($\boldsymbol{\eta}$s) act as offsets on the intercept.

Putting it all together, our model is defined mathematically as follows, where $i$ denotes observation, $j$ denotes plot, and $t$ denotes year. We assume the observations are conditionally Gaussian,

$$\mathbf{y} \sim \text{Normal}\left(\boldsymbol{\mu}, \sigma^2\right), \tag{A2.1}$$

where $\boldsymbol{\mu}$ is the determinstic expectation from the regression model,

$$\mu_i = \boldsymbol{\beta}\mathbf{x}_i + \boldsymbol{\gamma}_{j(i)}\mathbf{z}_i + \eta_t. \tag{A2.2}$$

All fixed effect $\beta$s were drawn from normally distributed priors with mean 0 and standard deviation 1: $\beta \sim \text{Normal}(0, 1)$. $\gamma$ random effects were drawn from a multivariate prior centered on zero with a shared variance covariance matrix:

$$\boldsymbol{\gamma} \sim \text{MVN}\left(0, \Sigma\right), \tag{A2.3}$$

$$\Sigma \sim \text{Cholesky...} \tag{A2.4}$$

\noindent{The random year effects ($\boldsymbol{\eta}$) are modeled as intercept offsets centered on zero with a shared variance ($\sigma_{yr}$): $\boldsymbol{\gamma} \sim \text{Normal}\left(0, \sigma_{yr}\right)$.

We fit the model using MCMC as implemented in the software Stan (Stan Development Team 2016). Our Stan code is below. All code necessary to reproduce our results has been archived on Figshare (*link here*) and released on GitHub (https://github.com/atredennick/usses_water/releases/v0.1).

## Section A2.1.2 Stan Code

```
data {
  int<lower=0> Npreds;          # number of covariates, including intercept
  int<lower=0> Npreds2;         # number of random effect covariates
  int<lower=0> Nplots;          # number of plots
  int<lower=0> Ntreats;         # number of treatments
  int<lower=0> Nobs;            # number of observations
  int<lower=0> Nppts;           # number of precip levels to predict
  int<lower=0> Nyears;          # number of years
  vector[Nobs] y;               # vector of observations
  row_vector[Npreds] x[Nobs];     # design matrix for fixed effects
  row_vector[Npreds2] z[Nobs]; # simple design matrix for random effects
  matrix[Nppts,Npreds] newx;    # design matrix for predictions
  matrix[Npreds2,Npreds2] R;      # priors for covariance matrix
  int plot_id[Nobs];            # vector of plot ids
  int treat_id[Nobs];           # vector of treatment ids
  int year_id[Nobs];            # vector of year ids
}

parameters {
    vector[Npreds] beta;                  # overall coefficients
    vector[Nyears] year_off;            # vector of year effects
    cholesky_factor_corr[Npreds2] L_u; # cholesky factor of plot random effect corr matrix
  vector[Npreds2] beta_plot[Nplots]; # plot level random effects
    vector<lower=0>[Npreds2] sigma_u;  # plot random effect std. dev.
    real<lower=0> sd_y;                # treatment-level observation std. dev.
    real<lower=0> sigma_year;          # year std. dev. hyperprior
}

transformed parameters {
  vector[Nobs] yhat;                # vector of expected values
  vector[Npreds2] u[Nplots];        # transformed plot random effects
  matrix[Npreds2,Npreds2] Sigma_u; # plot ranef cov matrix

  Sigma_u = diag_pre_multiply(sigma_u, L_u); # plot-level covariance matrix
  for(j in 1:Nplots)
    u[j] = Sigma_u * beta_plot[j]; # plot random intercepts and slopes
```

```
  # regression model for expected values (one for each plot-year)
  for (i in 1:Nobs)
    yhat[i] = x[i]*beta + z[i]*u[plot_id[i]] + year_off[year_id[i]];
}

model {
  ####  PRIORS
  sigma_year ~ cauchy(0,2)
  year_off ~ normal(0,sigma_year); # priors on year effects, shared variance
  beta ~ normal(0,1);                 # priors on treatment coefficients
  L_u ~ lkj_corr_cholesky(1);       # prior on the cholesky factor which controls the
                                    # correlation between plot level treatment effects

  for(i in 1:Nplots)
        beta_plot[i] ~ normal(0,1); # plot-level coefficients for intercept and slope

    ####  LIKELIHOOD
  for(i in 1:Nobs)
    y[i] ~ normal(yhat[i], sd_y); # observations vary normally around expected values
}
```

## Section A2.2 Figures




**Figure A2-1** Scatterplot, on the arithmetic scale, of the data and model estimates shown as solid lines. Model estimates come from treatment level coefficients (colored lines).




**Figure A2-2** Posterior distributions of random year effects (intercept offsets). Kernel bandwidths of posterior densities were adjusted by a factor of 5 for visual clarity.

# References

Stan Development Team. 2016. Stan: A C++ Library for Probability and Sampling, Version 2.14.1.